


В настоящее время существует большой набор дополнительных устройств для IBM совместимых компьютеров. Это позволяет создать оптимальную и удобную систему для работы и отдыха. Часто компьютер используется не совсем по своему прямому назначению, например, при прослушивании музыки с компакт-дисков. В этих случаях удобно управлять компьютером на расстоянии с помощью пульта дистанционного управления (ПДУ) на ИК лучах. Уже существуют устройства CD-ROM с ПДУ, что значительно повышает удобство работы с компьютером. Описываемое устройство позволяет ввести подобную возможность в компьютеры, в которых отсутствует система дистанционного управления. Также это устройство может использоваться в качестве джойстика в игровых программах или в качестве “мыши” в некоторых офисных приложениях.

Устройство ДУ, схема которого приведена на рис. 1, представляет собой приемник ИК сигналов от пульта дистанционного управления. Сигнал, принятый фотодиодом VD1, подается на вход усилителя-формирователя на микросхеме DA1. В этой микросхеме сигнал усиливается, фильтруется и преобразуется в сигнал TTL уровня. С выхода этой микросхемы сигнал подается на элемент DD1.1, после которого поступает на вывод C3 ISA-шины компьютера.





При нажатии кнопки на ПДУ вырабатывается сигнал, который после обработки микросхемой DA1 подается на вход IRQ10 ISA шины. По каждому нарастанию сигнала вырабатывается сигнал аппаратного прерывания с номером 10h. По этому сигналу процессор вызывает

подпрограмму, адрес которой указан в векторе прерывания с номером 72h. Длительности импульса, вырабатываемого ПДУ, достаточно для перевода процессора в режим обработки прерывания, поэтому не требуется дополнительно определять длительность действия сигнала на вход IRQ10. После выполнения подпрограммы процессор возвращается к выполнению основной программы. Информацию о нажатой кнопке могут использовать другие программы, в том числе те, которые управляют другими устройствами, для выполнения необходимых действий.

В качестве ПДУ можно использовать, например, пульт промышленного изготовления от телевизоров ЗУСЦТ, собранный на микросхеме КР1506ХЛ1, описанный в [1]. Этот пульт формирует последовательность из 14 импульсов, длительность пауз, между которыми определяет передаваемую команду. Пример программы-драйвера для DOS, приведенный в табл.1, являясь резидентным обработчиком прерывания 72h, обрабатывает эти последовательности и предоставляет через прерывание 2Fh информацию о нажатой кнопке. Эту программу нужно загрузить в начале работы компьютера, при этом она остается в памяти резидентно. Для того, чтобы узнать было ли нажатие кнопки и определить код этой кнопки, вызывают прерывание 2Fh, поместив в регистр AX значение 4E01h. После этого в регистре DX будет содержаться код нажатой кнопки, а в регистре AL статус: 00-после последнего запроса ни одна кнопка не была нажата, 01-кнопка была нажата. Поместив в AX число 4E00h и вызвав прерывание 2Fh, можно узнать, загружен ли драйвер в память. Если после вызова прерывания в регистре AL содержится FFh, то драйвер загружен. Подробнее о командах пульта можно прочитать в [1]. Число, являющееся константой периода сигнала с ПДУ, вычисляется подпрограммой, в зависимости от быстродействия компьютера, но может быть определена экспериментально для каждого типа компьютера и пульта. Подпрограмма не приводится в связи с ее достаточно большим размером.

Устройство смонтировано на 2 печатных платах. На первой установлены фотодиод VD1, микросхема DA1, конденсаторы C1...C4 и

резистор R1. Эта плата закрыта экраном из меди или латуни, соединенным с общим проводом. Она размещена на передней панели системного блока. Напротив фотодиода VD1 должно быть отверстие, прикрытое светофильтром красного цвета. Способ крепления этой платы может быть произвольным, в зависимости от конструкции корпуса компьютера. На второй плате размещена микросхема DD1 и конденсаторы C5...C6. Эта плата вставляется в дополнительную (короткую) часть свободного ISA слота на системной плате. Чертежи печатных плат приведены на рис.2. Соответствующие выводы плат с номерами 1-3 необходимо соединить между собой с помощью ленточного кабеля. Можно также использовать 3 провода, переплетенные между собой.



В устройстве использован постоянный резистор МЛТ, конденсаторы C2, C6 типа К50-35, остальные любые малогабаритные. В качестве микросхемы DA1 можно использовать КР1084УИ1 или импортную TBA2800. На месте DD1 можно установить К555ЛА8 или

другой логический элемент с инверсией и открытым коллектором на выходе.

Правильно собранное устройство начинает работать сразу. Иногда бывает необходимо подобрать конденсатор С2 по максимальной дальности работы устройства.

В качестве ПДУ можно использовать и другие пульты, в том числе и импортные с соответствующим изменением программы-драйвера. Описанное устройство может быть использовано под операционными системами OS/2 и WINDOWS при написании соответствующих программ-драйверов.

Литература.

1. Ельяшкевич С. Цветные стационарные телевизоры и их ремонт.- М.: Радио и связь, 1990, с. 27.

Приложение 1.

;Пример программы-драйвера RCU.COM

```
.model tiny
.code
.386
org      100h

body:    jmp init                ;Переход на инициализацию

old      db 4 dup (0)           ;Старый адрес вектора 2Fh
buf      dw 15 dup (0)         ;Буфер
nc       dw 0                  ;Номер текущего импульса
tc       dw 0                  ;Последнее значение таймера
pe       db 0                  ;Флаг готовности

tim      equ 400                ;"Граница" между "0" и "1"
max      equ 5000              ;"Граница" между "1" и паузой

int2Fh:  ;Обработчик прерывания 2Fh
        cmp ax,0E400h          ;Для проверки наличия
        jne p1                 ;RCU.COM вызвать 2Fh при
        mov ax,0E4FFh         ;AX=0E400h. Возвращает
        iret                   ;AX=0E4FFh

p1:      cmp ax,0E401h          ;Опрос состояния при
        jne p4                 ;AX=0E401h
        push ds si bx          ;Сохранить регистры
        push cs
        pop ds                 ;В DS:SI
        lea si,buf             ;начало буфера
        xor dx,dx              ;DX=0
```

```

p2:    mov bx,1000h      ;BX=2^12
        lodsw           ;AX=[DS:SI], SI=SI+2
        cmp ax,tim      ;Сравнить с периодом
        jl p3           ;Перейти, если меньше
        add dx,bx       ;Увеличить DX на BX
p3:    shr bx,1         ;BX=BX/2
        cmp bx,0        ;Продолжить,
        jne p2          ;если не нуль в BX
        pop bx si ds    ;Восстановить регистры
        mov al,cs:[pe]  ;В AL вернуть флаг готовности
        mov cs:[pe],0   ;Сбросить флаг
        iret            ;Вернуться
p4:    jmp dword ptr cs:[old] ;Перейти на старый обработчик

int72h:
        pusha           ;Обработчик прерывания 72h
        pushf           ;Сохранить регистры
        xor al,al       ;и флаги
        out 43h,al      ;AL=0
        jmp $+2         ;Зафиксировать значение
        in al,40h       ;системного таймера
        jmp $+2         ;Прочитать младший байт
        mov ah,al       ;AH=AL
        in al,40h       ;Прочитать старший байт
        xchg ah,al      ;Поменять местами AH и AL
        mov dx,cs:[tc] ;В DX старое значение таймера
        mov cs:[tc],ax  ;Сохранить новое значение
        sub dx,ax       ;В DX интервал между вызовами
        cmp dx,max      ;Если интервал более max, то
        jb p5           ;произошла ошибка или пауза
        mov cs:[nc],0   ;между пакетами импульсов,
        jmp p6          ;обнуляем номер импульса
p5:    cmp cs:[pe],1    ;Если готовность (информация
        je p6           ;считана), то выйти
        mov bx,cs:[nc]  ;BX=номер текущего импульса
        push bx         ;Сохранить BX
        lea si,buf      ;В CS:SI начало буфера
        shl bx,1        ;BX=BX*2
        mov cs:[si+bx],dx ;Поместить в буфер значение
        inc cs:[nc]     ;Увеличить номер импульса
        pop bx          ;Восстановить BX
        cmp bx,12       ;Если не последний импульс,
        jb p6           ;то выйти
        mov cs:[pe],1   ;Установить флаг готовности
        mov cs:[nc],0   ;Обнулить номер импульса
p6:    mov al,20h       ;Послать контроллерам
        out 20h,al      ;прерываний сигнал
        jmp $+2         ;о завершении прерывания
        out 0A0h,al
        sti             ;Разрешить прерывания
        popf            ;Восстановить флаги
        rora           ;и регистры
        iret            ;Вернуться

init:
        push cs         ;Инициализация
        pop ds          ;Загрузить DS
        mov ah,09h      ;Вывести
        lea dx,msg1     ;строку
        int 21h

```

```
mov ax,352Fh      ;Узнать вектор
int 21h          ;прерывания 2Fh
mov word ptr old,bx ;и сохранить
mov word ptr old+2,es ;его в OLD
mov ax,252Fh     ;Установить
mov dx,offset int2Fh ;вектор
int 21h          ;прерывания 2Fh
mov ax,2572h     ;Установить вектор
mov dx,offset int72h ;прерывания 72h
int 21h          ;(обработчик IRQ10)
in al,21h        ;Разрешаем
and al,0FBh      ;прерывание
out 21h,al       ;IRQ2
jmp $+2
jmp $+2
in al,0A1h       ;Разрешаем
and al,0FBh      ;прерывание
out 0A1h,al      ;IRQ10
mov dx,offset init ;Вычислить размер
shr dx,4         ;программы
add dx,16        ;и оставить ее
mov ax,3100h     ;резидентно
int 21h

msg1: db 'Драйвер системы дистанционного управления '
      db 'на ИК лучах.',13,10
      db 'Установлен на IRQ 10.',13,10,'$'
      end body
```